木構造を処理するマクロの作成－SQL プロシジャを利用して－

堀田真一

株式会社シーエーシー

医薬システムビジネスユニット

The Creation of macro to treat the tree structure: Using SQL procedure

Shinichi Hotta

Medical & Pharmaceutical System Business Unit, CAC Corporation
本資料はSAS Forum ユーザー会　学術総会 2005における口頭発表「木構造を処理するマクロの作成－SQL プロシジャを利用して－」の中で発表したサンプルコードの詳細を示したものです。

　本資料で紹介されたサンプルコード、または、それを改変したものを使用することは自由です。しかし、使用して発生した結果について、作者は一切の責任を負いません。

　本資料に関するお問合せは下記までお願いします。

〒101-0006

東京都中央区日本橋富沢町１０番１６号

TEL.(03)3664-0920 　FAX.(03)3664-1076

E-mail shotta@ZIPANGU.cac.co.jp
	名前
	説明

	indata1
	親子関係を表現したデータセット（スライド7、10）。

	indata2
	水準名の順序を表現したデータセット（スライド7、10）。

	sample
	水準名と値を格納したデータセット（スライド10）。


表1　データセット一覧

表2　サンプルコード一覧
	番号
	説明

	1
	部分集団への分割を行うマクロ（スライド8）。水準の数が多くなると、多大なシステムリソースを費やす可能性があるので注意。

	2
	サンプルコード1を実行した結果作成されるデータセットとデータセットindata2を使って、データセットsampleに対し、部分集団名（変数subgrp）と部分集団番号（変数subgrpid）と部分集団への分割パターンの表示（変数model）でラベル付けを行うプログラム例（スライド10）。

	3
	サンプルコード2を実行した結果作成されるデータセットから、ダミー変数によりモデルが表現したデータセット（スライド11）を作成するプログラム例（スライド10）。

	4
	サンプルコード3を実行した結果作成されるデータセットを使って、各々の分割パターンに対して、REGプロシジャによりAICを計算するプログラム例（スライド11）。


************************************************************************************************ 

サンプルデータセット sample

変数の説明

 level 水準名

 y     観測値

SAS version : 8.2

************************************************************************************************ ;

data sample ;

length level $8 ;

  do i=1 to 30 ;

    y=normal(999)*2+4 ;


level="level1" ;


output ;

  end ;

  do i=1 to 33 ;

    y=normal(999)*2 ;


level="level2" ;


output ;

  end ;

  do i=1 to 28 ;

    y=normal(999)*2 ;


level="level3" ;


output ;

  end ;

run ;

************************************************************************************************ 

サンプルデータセット indata1

変数の説明

 pr 親子関係の親を表した変数

 ch 親子関係の子を表した変数

SAS version : 8.2

************************************************************************************************ ;

data indata1 ;

  pr=. ;

  ch=1 ;

  output ;

  do pr=1 to 3 ;

    do ch=1 to pr+1 ;

      output ;

    end ;

  end ;

run ;

************************************************************************************************ 

サンプルデータセット indata2

変数の説明

 level 水準名

 id    水準名が持つ通し番号

SAS version : 8.2

************************************************************************************************ ;

data indata2 ;

  length level $8 ;

  do id=1 to 3 ;

    level=compress('level'||id) ;

    output ;

  end ;

run ;

************************************************************************************************ 

サンプルコード1

機能　部分集団への分割パターンを作成するマクロ

マクロ変数の説明

 indata1　親子関係の記述を格納した入力データセット名の指定

 indata2　水準名とその順序を格納した入力データセット名の指定

 outdata　出力データセット名の指定

 level　　水準名を格納した変数名の指定

 id       水準名の通し番号を格納した変数名の指定

 tmp      一時的なバッファとして使用するデータセット名の指定

出力データセット内の変数の説明

 depth  処理された水準の数

 divide 部分集団の数

 dest   部分集団への分割パターンの数字による表現

 model  部分集団への分割パターンの文字列による表現

注意　水準の数が多くなると、多大なシステムリソースを費やす可能性があります。

SAS version : 8.2

************************************************************************************************ ;

%macro divide(indata1=,indata2=,outdata=,level=level,id=id,tmp=tmp);

proc sql noprint ;

create table &outdata (depth num, divide num, dest char(200), model char(200), pr num, ch num) ;

create table &tmp as

select 1 as depth,

       1 as divide,

       compress(put(struc.ch,best.)) as dest,

       compbl(ord.&level) as model,

       struc.pr,

       struc.ch 

from &indata1 struc, &indata2 ord

where pr=. and calculated depth=ord.id ;

%do %while(&SQLOBS) ;

create table &outdata as select * from &outdata union all select * from &tmp ;

create table &tmp as

select pr.depth+1 as depth,

       max(ch.pr,ch.ch) as divide,

       compbl(pr.dest||put(ch.ch,best.)) as dest,

       case when ch.pr<ch.ch then 

       compbl(pr.model||" "||compress(ord.&level))

       else 

       tranwrd(pr.model,scan(pr.model,ch.ch),compress(scan(pr.model,ch.ch)||"="||compress(ord.&level)))

       end 

       as model,

       ch.pr,

       ch.ch 

from &tmp pr, &indata1 ch, &indata2 ord

where pr.divide=ch.pr and pr.depth+1=ord.id ;

%end ;

drop table &tmp ;

quit ;

%mend ;

%divide(indata1=indata1,indata2=indata2,outdata=outdata);

************************************************************************************************ 

サンプルコード2

機能　サンプルコード1を実行した結果作成されるデータセット（データセット名をoutdataとしている）

     とデータセットindata2を使って、データセットsampleに対し、部分集団名（変数subgrp）と部分

     集団番号（変数subgrpid）と部分集団への分割パターンの表示（変数model）でラベル付けを行い、

     データセットmodeldtを作成する。水準の数を3としている。

出力データセット内の変数の説明

 level    水準名

 y        観測値

 model    部分集団への分割パターンの文字列による表現

 subgrp   水準が所属する部分集団の文字列による表現

 subgrpid 水準が所属する部分集団の、分割パターン内での通し番号

SAS version : 8.2

************************************************************************************************ ;

proc sql ;

create table modeldt as 

select outdt.model,

       scan(outdt.model,input(scan(outdt.dest,ord.id),best.),' ') as subgrp, 

       input(scan(outdt.dest,ord.id),best.) as subgrpid,

       samp.*

from outdata outdt, sample samp, indata2 ord

where outdt.depth=3 and samp.level=ord.level ;

quit ;

************************************************************************************************ 

サンプルコード3

機能　サンプルコード2を実行した結果作成されるデータセット（データセット名をmodeldtとしている）

     から、ダミー変数（変数col1、col2、col3）により分割パターン（変数model）を表現したデータ

     セットmddesignを作成する。水準の数を3としている。

出力データセット内の変数の説明

 y        観測値

 model    部分集団への分割パターンの文字列による表現

 col1     オブザベーションが所属する部分集団を表すダミー変数

 col2     オブザベーションが所属する部分集団を表すダミー変数

 col3     オブザベーションが所属する部分集団を表すダミー変数

SAS version : 8.2

************************************************************************************************ ;

proc sort data=modeldt ; by model ; run ;

proc glmmod data=modeldt outdesign=mddesign noprint ;

class subgrpid ;

by model ;

model y=subgrpid/noint ;

run ;

data mddesign ;

  set mddesign ;

  array cols{3} col1-col3 ;

  do i=1 to 3 ;

    if cols{i}=. then cols{i}=0 ;

  end ;

run ;

************************************************************************************************ 

サンプルコード4

機能　サンプルコード3を実行した結果作成されるデータセット（データセット名をmddesignとしている）

     を使って、各々の分割パターン（変数model）に対して、REGプロシジャによりAICを計算する。

SAS version : 8.2

************************************************************************************************ ;

proc reg data=mddesign outest=AIC noprint ;

model y=col: /aic rmse ;

by model ;

run ;

quit ;

proc sort data=AIC ; by _aic_ ;run ;
